
AN82250は、PSoC® 3、PSoC 4、PSoC 5LPおよび PSoC 6の PLD部にプログラマブル デジタル ロジックの設計

を実施する方法を説明します。また、PSoCの汎用デジタルブロック (UDB) およびそれらのプログラマブルロジック

デバイス (PLD) のサブブロックを紹介します。サンプルプロジェクトは、PSoC Creator™内に Verilogベースのコン

ポーネントを作成することにより、設計で PLDを使用する方法を説明します。 

1 はじめに 

PSoC 3、PSoC 4、PSoC 5LPおよびPSoC 6 (以降はPSoCと記載) は単なるマイクロコントローラ以上のものです。

PSoCにより、マイクロコントローラ、コンプレックス プログラマブル ロジック デバイス(CPLD)、および高性能ア

ナログの機能を統合し、他に例を見ない柔軟性を得ることができます。これにより、費用、基板面積、消費電力、お

よび開発時間を低減できます。 

注：このアプリケーションノートは、UDBがないデバイスには適用されません。デバイスに UDBがあるかどうかは、

そのデバイスのデータシートを参照してください。 

このアプリケーションノートは、PSoCの汎用デジタルブロック (UDB) 内の PLDを紹介し、PSoC Creatorコンポー

ネントを作成することにより、それらを使用する方法を説明します。それは、コンプレックス プログラマブル ロ

ジック デバイス (CPLD) 機能を PSoCに移植するための効果的な最初のステップです。このアプリケーションノート

をお読みいただいた後は、PSoC PLDに詳しくなり、PSoC Creatorを使用して Verilogベースの独自のカスタム コ

ンポーネントを作成することができます。 

PSoCのデジタル機能の詳しい活用方法については、次のステップとして「AN82156 – PSoC 3, PSoC 4 and PSoC 

5LP Designing PSoC Creator Components with UDB Datapaths」をお読みください。 

http://www.cypress.com/
http://www.cypress.com/index.cfm?rID=67774
http://www.cypress.com/?rID=69774
http://www.cypress.com/?rID=xxxx&source=anxxxxx
http://www.cypress.com/?rID=69774
http://www.cypress.com/?rID=69774

これは上級のアプリケーションノートであり、読者が PSoCおよび PSoC Creatorに精通していることを前提にして

います。PSoCが初めての場合は、下記をまず参照してください。 

▪ AN54181 Getting Started with PSoC 3  

▪ AN79953 Getting Started with PSoC 4  

▪ AN77759 Getting Started with PSoC 5LP  

▪ AN221774 Getting Started with PSoC 6 MCUs 

PSoC Creatorが初めての場合は、下記を参照してください。 

▪ PSoC Creator home page 

注： 

このアプリケーションノートは、デジタル設計および Verilog の基礎的な知識があることを前提にしています。これ

らの概念が初めてであれば、「AN81623 – PSoC 3 and PSoC 5LP Digital Design Best Practices」および「KBA86336 

– Just Enough Verilog for PSoC」を参照してください。関連の PSoCデジタル設計のリソースは「参考資料」節に記

述されています。 

2 PSoC UDB 

PSoC は、汎用デジタルブロック (UDB) と呼ばれる小型、高速、低電力デジタル ブロックのアレイによってプログ

ラマブル ロジックを実装しています。PSoCのデバイスは最大 24個の UDBがあります。図 1に示すように、UDB

は 2個の小型プログラマブル ロジック デバイス (PLD)、1個のデータパス モジュール、および 1個のステータス コ

ントロール ロジックから成っています。 

図 1. UDBのブロック図 

 

名前が示す通り、プログラマブル ロジックは、論理素子 (AND、OR、INVERT および FLIP-FLOP) のアレイを備え

ているデバイス ファミリーです。一般的に、PLDは、特定のロジック機能を実行するよう設定できる回路です。 

PSoC PLD は、レジスタ付きまたは組み合わせ積和ロジック、ルックアップ テーブル、マルチプレクサ、ステート 

マシン、およびデータパス動作の制御などを作成するのに使用されます。UDB データパスの詳細については、

AN82156を参照してください。 

http://www.cypress.com/documentation/application-notes/an221774-getting-started-psoc-6-mcu
http://www.cypress.com/go/creator
http://www.cypress.com/?rID=67774
http://www.cypress.com/?id=4&rID=76925

2.1 PSoC UDB内の PLDのアーキテクチャ 

PSoC PLDは、ほとんどの標準 PLDと同様、ANDアレイの後に ORアレイが続きます。それらの両方はプログラム

可能です。これは、普通、積和アーキテクチャと呼ばれています。 

ANDアレイ内の 8本のプロダクト ターム (PT) に 12本の入力を供給します。各 PT内で、入力の真 (T) またはその補

数 (C) を選択できます。PTの出力は ORアレイへの入力となります。ORゲートの出力はマクロセル (MC) に提供さ

れます。マクロセルは、追加の組み合わせロジックを持ったフリップ フロップです。 

図 2に示すように、各 UDBに 2個の PLDがあり、それぞれには 8本の PTと 4個のマクロセルがあります。PSoC

には最大 48個の PLDがあり、そのため最大 192個のマクロセルと最大 384本の PTがあります。各 PLDは独立し

ています。また、キャリーチェーンを介して接続するか、またはデジタル システム インターコネクト (DSI) に接続

することができます。 

付録 Aでは、PSoC PLDリソースを競合他社の同じサイズの PLDと比較します。 

図 2. PSoC PLDの構造 

 
 

図 3には、PSoC PLDにマップされた論理式の例を示します。 

図 3. PLDにマップされた論理式 

 

 
マクロセルのアーキテクチャは図 4に示されます。マクロセル出力はレジスタ付き出力、または組み合わせ出力のい

ずれかです。付録 Bでは、2つの例を使ってマクロセルを介したデータフローを説明します。詳細情報については、

使用しているデバイスのテクニカル リファレンス マニュアルの UDB章のマクロセルの節を参照してください。 

図 4. PSoC PLDマクロセルのアーキテクチャ 

3 PSoC Creator 

PSoC Creator は、ハードウェア開発に対応した回路図ベースの環境を提供します。それは、次の 2 つの大まかな方

法で UDB PLD内にロジック機能とステート マシンを実装することができます。 

1. Verilog: PSoC Creatorは、ハードウェア記述言語 (HDL) である Verilogに対応しています。Verilogを使用して、

PSoC UDB にマップされるデジタル機能を実装できます。このプロセスでは、PSoC Creator に備えられた

VerilogコンパイラであるWarp™合成ツールを使用します。 

このアプリケーションノートは、Verilogベースのコンポーネントの作成方法を説明します (図 5を参照してくだ

さい)。 

Verilogの詳細については、「KBA86336 – Just Enough Verilog for PSoC」を参照してください。 

注：Warpの詳細については、PSoC CreatorのWarp Verilogリファレンス ガイド (Help > Documentation) を参

照してください。 

図 5. Verilogベースのコンポーネント 

 

2. 回路図: このプロセスは、必要な機能を実行するために個別のゲート (AND、OR、XOR、NOT)、DFF、および

他のデジタル論理ブロックを配線することを必要とします。PSoC Creator は、すべての論理演算だけでなく、

マルチプレクサ、ルックアップ テーブル (LUT)、および他の簡単な PLD ベースの機能にもゲート記号を提供し

ます。 

また、PSoC Creator は作成済み、かつテスト済みの標準ペリフェラル コンポーネントも提供します。これらの

コンポーネントは、PLDとデータパスの両方を含む UDBアレイにマップされます。その一部を図 6に示します。

これらのコンポーネントを使用するのは、Verilogを使わずに PSoCの PLD機能を使用する際の最も簡単な方法

です。 

http://www.cypress.com/
http://www.cypress.com/?id=4&rID=76925

図 6. PSoC Creator内のデジタル コンポーネント 

 

 

4 サンプル プロジェクト 

PSoCについて学ぶための最良の方法の 1つは、それを使用することです。このサンプル プロジェクトでは、簡単な

PLDベースの Verilogコンポーネントを作成する手順について説明します。 

まず、アプリケーションノートのランディング ページで AN82250.zipファイルをダウンロードします。プロジェクトを

見るために、そのファイルをフォルダに解凍して、PSoC Creatorで AN82250.cywrkファイルを開きます。このプロ

ジェクトは、回路図面の指示に従って CY8CKIT-001 PSoC開発キット (DVK) の PSoC 5LPで作業するよう設計しま

した。少し変更すると、他の開発プラットフォームで実行できます。このプロジェクトをビルドして PSoC DVK に

プログラムします。 

このサンプル プロジェクトでは、ファームウェアを必要とせず、5ビットのシーケンス検出器をハードウェアに完全

に実装します。回路図面の詳細については、図 7を参照してください。このプロジェクトの重要な特長は、回路図面

に表示されている全てのコンポーネントが、クロックとピンを除き、UDB PLDに実装されていることです。 

このプロジェクトは 2進数のパターンを入力とします。パターンは、PSoC DVKでの 2個の押しボタン式スイッチに

より生成されます。「SW_1」スイッチのボタンを押すと論理 0、また「SW_2」スイッチのボタンを押すと論理 1と

解釈されます。検出器の状態を示すために、4個の出力で、DVK上の LEDを駆動します。 

PSoCをリセットすると、LEDが灯り、PSoCが入力受け入れ準備 (即ち、スイッチを押すこと) ができたことを示し

ます。PSoC は図 8 に示した状態図に従います。不完全ではあるが、正しいシーケンスを入力すると、LED は消え

て、部分的なシーケンスが入力されたことを示します。間違えてスイッチを押すと、4個の LEDが灯ります。完全な

5ビット シーケンスを正しく入力すると、LEDは 2進数でカウントし始めます。 

http://www.cypress.com/
http://www.cypress.com/?rID=69773
http://www.cypress.com/?rID=37464

図 7. サンプル プロジェクト用の TopDesign回路図面 

 

入力から出力への信号の流れは以下の通りです。 

▪ 2個の押しボタン スイッチの入力は、デバウンサ コンポーネントを使って跳ね返りが抑えられ、エッジが検出さ

れます。このコンポーネントをアクセスするためには、PSoC Creatorをコンポーネント パック 4かそれ以降のもの

に更新してください。 

▪ 入力ピンは、抵抗プルアップとして設定されます。そのため、押しボタンの入力は、スイッチを押すとハイから
ローに変化します。従って、デバウンサの「ネガティブ エッジでの検出」出力は、スイッチが押されたことが有

効であることを示すために使用されます。 

▪ その後、これらの信号は、10110のシーケンスを検出するよう設定された SeqDetectorコンポーネントに行きま

す。このパターンは、コンポーネントのカスタマイザで 0 (00000) ～ 31 (11111) の間の値を入力することで変更

できます (図 9を参照してください)。 

▪ シーケンス全体を正しく入力すると、「検出」出力がアサートされます。入力をたった一度だけ間違えても、
「再起動」出力がアサートされます。 

▪ 「検出」信号がアサートされると、4ビット カウンターはカウントし始めます。それまでは、リセット状態が維

持されます。カウンター周期は、コンポーネントのカスタマイザ内に 1～15の間の望ましい 4ビット周期値を入

力することで調整できます (図 10を参照してください)。 

▪ 「再起動」および「検出」信号は出力のマルチプレクサを制御して、図 8の状態図に基づき 4個の LEDを駆動し

ます。 

http://www.cypress.com/
http://www.cypress.com/?rID=56745

図 8. サンプル プロジェクトの状態図 

 

 

図 9. SeqDetectorコンポーネントのカスタマイザ 

図 10. カウンター コンポーネントのカスタマイザ 

 

 

注: 4 ビット カウンター用の Verilog ファイルとシーケンス検出器コンポーネントを表示するために、Workspace 

Explorerの Componentsタブに移動してください。 

PSoC PLDを効果的に使用するための重要な点は、PSoC Creatorで Verilogベースのコンポーネントを作成すること

です。「KBA86338 – Creating a Verilog-based Component」では、Verilogベースのコンポーネントの作成プロセス

をまとめます。事例として SeqDetectorと Counter4Bitコンポーネントを使用すると、このプロセスについて理解で

きるようになります。 

4.1 Verilogコンポーネントの作成: Counter4Bit 

最も簡単なカスタム Verilogベースのコンポーネントの 1つは、同期リセットとイネーブル機能を備えた 4ビットの

アップカウンターです。 

4.1.1 4ビット カウンター コンポーネントの作成手順 

既存のプロジェクトに新しいコンポーネントを追加することができますが、この事例では、既存のプロジェクトが無

い状態から始めます。 

1. PSoC Creatorを起動して、新規プロジェクトを作成します。「MyComponents」をプロジェクト名とします。 

2. Workspace Explorer の Source タブで、MyComponents ワークスペースを右クリックして、Add > New 

Projectをクリックします。 

3. この新規プロジェクトをコンポーネント ライブラリに設定するために、Create Project ダイアログ ボックスで

Design project の下にある PSoC Library を選択し、Next をクリックします。次の画面で、このライブラリを

作るための CPU コアを選択します。その後、ロケーションをデフォルトのままにして、Project name を

「MyLibrary」に変更します。 

作成したライブラリに対し、新しいコンポーネントを以下の通りに追加します。 

4. Components タブで、「MyLibrary」プロジェクトを右クリックして、コンテキスト  メニューから Add 

Component Itemをクリックします(図 11を参照してください)。 

バージョン番号をコンポーネント名に含めるのは良いやり方です。コンポーネント名に「_vX_Y」タグを加えま

す (その内、「X」はメジャーバージョンであり、「Y」はマイナー バージョンです)。PSoC Creatorはバージョ

ン管理機能を持っていて、コンポーネントの複数のバージョンを追跡できます。 

5. Symbol Wizard コンポーネントのテンプレートを選択して、コンポーネントに「Count4Bit_v1_00」と名付けま

す。 

図 11. カスタム コンポーネントを作成する 

 

空のシンボルから始めることができますが、この事例では、時間をかけないためにウィザードを使用します。詳

細については、Help > Documentationにあるコンポーネント作成者ガイドを参照してください。 

6. Create Newボタンをクリックして、コンポーネントのシンボル ウィザードを起動します。 

このウィザードでは、入力と出力を定義するよう要求して、この情報を使ってコンポーネント シンボルを作成

します。 

7. 図 12に示すように、回路図シンボル用に 3個の入力端子と 2個の出力端子を定義します。 

図 12. Count4Bit用のシンボル作成ウィザード 

 

図 13に示すように、OKをクリックして、シンボル回路図でシンボルを生成します。 

図 13. 4ビット カウンター用の最初のシンボル 

 

図 14に示すように、コンポーネントのサイズを変更でき、コンポーネントの外観も変更できます。 

図 14. 4ビット カウンターの最終シンボル 

 

8. シンボル回路図で空きスペースを右クリックして、Properties をクリックします。図 15 に示すように、プロパ

ティ フィールドの Symbolセクションで、Doc.CatalogPlacementの省略記号 (…) をクリックします。 

図 15. シンボル属性のダイアログボックス 

 

9. 図 16に示すように、Catalog Placementダイアログで、「Community/Digital/Logic/Counter 4-bit」を入力します。 

これにより、カウンターは、Component Catalog ウィンドウの Community タブで、「Digital」フォルダの

「Logic」サブフォルダの中に配置されます。そのカタログ名は「Counter 4-bit」です。 

図 16. カタログ配置を設定する 

 

カウンター用に設定可能な周期値を持たせるために、コンポーネントのパラメーターを追加する必要があります。 

10. シンボル回路図で空きスペースを右クリックして、Symbol Parametersをクリックします (図 17を参照してく

ださい)。 

パラメーターの名、タイプ、およびデフォルトの値を、それぞれ「period」、「uint8」、および「15」に指定し

ます。このパラメーターにより、ユーザーはカスタマイザでカウンターの周期値を指定できます (図 10を参照し

てください)。 

図 17. Count4Bitのシンボル パラメーター 

 

11. Parameter Definitionダイアログ ボックスのMiscセクションで Descriptionフィールドをクリックして、この

パラメーター用に記述します。 

Validatorsフィールドをクリックして、パラメーターの検証ツールを設定します。検証ツールは、パラメーター

が許容入力範囲以内にあるかをチェックします。 

図 18のように、周期値が 1～15の間にあることを保証するよう検証ツールを設定します。OKをクリックして、

変更を適用します。 

図 18. Count4Bitの検証ツールを追加する 

 

 
12. 図 19に示すように、Parameter Definitionダイアログ ボックスで、Hardwareフィールドを Trueにセットしま

す。これは、パラメーターを Verilogファイルに渡すために必要です。 

図 19. パラメーターのハードウェアへの引き渡し 

 

次のステップとして、回路図シンボルを Verilogファイルにリンクさせます。PSoC Creatorは、コンポーネント 

シンボルに基づいて Verilogシェルを生成します。 

13. このためには、シンボル回路図で空きスペースを右クリックして、Generate Verilog をクリックします。図 20

のように、Generate Verilogダイアログ ボックスでデフォルト設定のままにして、Generateをクリックします。 

図 20. シンボル用の Verilogファイルの生成 

 

Target の値は、特定のデバイスの設定を制限するのに使用できますが、この事例ではデフォルト設定を使用します。 

作成したシンボルの Verilogファイルが出来上がります。 

注: Verilogファイルには、3組の#start header - #endがあります。このファイルを編集する時、全てのコードを

これらのセクション内に配置してください。これらのセクション外での Verilogファイルへの変更は、Verilogファイ

ルを再生成すると無くなります。 

この時点で、Verilogでカウンターを記述できるようになります。参照のために、完全なコードはコード 1に示します。 

4.1.2 Verilog設計: 4ビット カウンター 

まず、出力をレジスタ付き出力にします。Count4Bit_v1_00モジュールの I/Oリストを次のように変更します。 

 output reg [3:0] count, 

 output reg tc,  

注: Verilogファイルを再生成する場合、これらの変更を再び行う必要があります。また、これらの定義はファイル内

のどこか他の所で行うことはできません。 

次に、これが同期設計であるため、Verilogファイルで#start body と#endコメントの間に alwaysブロック (およ

びクロック エッジの情報) を加えます。 

 always @ (posedge clock)  

begin  

. . .  

 end 

注: タイミングと同期化の失敗の可能性を低減するために、PSoC設計でポジティブ エッジ クロックを使用したほう

が良いです。 

カウンターには同期リセット信号があります。この信号がアサートされると、「tc」と「count」の両方をクリアし

ます。 

 if(reset)      

begin 

   count <= 4'b0000; 

   tc <= 1'b0; 

 end  

注: 非同期リセット／プリセット信号もサポートされます。非同期フリップ  フロップ合成の詳細については、

「Warp Verilog Reference Guide」の 3.3.2節を参照してください。 

en 入力は、ハードウェア イネーブルです。この入力が LOW の場合、出力はアクティブのままであるが、コンポー

ネントの状態は変わりません。 

if(en)  /* start counting */ 

begin 

. . .  

end 

else   /* preserve state */ 

begin 

  count <= count; 

  tc <= tc; 

end 

count が周期値に達すると、count が period に等しいである限り、端末のカウント出力 tc は論理 1となる必要

があります。 

if(count == period)  

 begin 

   tc <= 1'b1; 

   count <= 4'b0000; 

http://www.cypress.com/


 

PSoC Creatorの Verilogによるプログラマブル ロジック設計の実装方法 

www.cypress.com 文書番号: 001-89367 Rev. *E 15 

 end 

そうでないと、4ビット カウンターは 0から period までカウントし、ポジティブ エッジごとに count出力をイン

クリメントする必要があります。 

else 

 begin 

   count <= count + 1; 

   tc <= 1'b0; 

 end 

Verilogファイルへの変更が終わったら、保存します。ここで、4ビット アップ カウンターの Verilog記述が完了しま

す。完全なコードはコード 1に示します。 

コード 1. 4ビット カウンターの完全な Verilog設計 

module Count4Bit_v1_00 ( 

  output reg [3:0] count, 

  output reg tc, 

  input   clock, 

  input   en, 

  input   reset 

); 

  

  parameter period = 0; 

 

//`#start body` -- edit after this line, do not edit this line 

 

  always @ (posedge clock) 

  begin 

    if(reset) 

    begin 

      count <= 4'b0000; 

      tc <= 1'b0; 

    end 

    else      

    begin 

      if(en) 

      begin 

        if(count == period) 

        begin 

          tc <= 1'b1; 

          count <= 4'b0000; 

        end 

        else 

        begin 

          count <= count + 1; 

          tc <= 1'b0; 

        end 

      end 

      else 

      begin 

        count <= count; 

        tc <= tc; 

      end 

    end 

  end 

 

//`#end` -- edit above this line, do not edit this line 

endmodule 

「MyComponents」プロジェクトでこのコンポーネントを使用するには、「MyLibrary」を依存関係として設定する

必要があります。 

そうするためには、SourceタブでMyComponentsを右クリックして、Dependenciesを選択します。図 21に示す

ように、User Dependenciesの下にある「MyLibrary」のチェックボックスがチェックされていることを確認してく

ださい。 

図 21. コンポーネントのライブラリ依存関係を追加 

 

設計で使用するためには、Component Catalog プロジェクトの TopDesign.cysch ファイルを開いて、Component 

Catalogに移動します。4ビット カウンターは Communityタブに配置されています。それを回路図にドラッグしド

ロップして、必要な接続を行います。 

注：ライブラリ プロジェクトの作成と使用方法については、PSoC Creator のヘルプ記事「Library Component 

Project」と「Basic Hierarchical Design Tutorial」を参照してください。 

次は、Verilogでシーケンス検出器コンポーネントを作成します。 

http://www.cypress.com/


 

PSoC Creatorの Verilogによるプログラマブル ロジック設計の実装方法 

www.cypress.com 文書番号: 001-89367 Rev. *E 17 

4.2 Verilogコンポーネントの作成: SeqDetector 

SeqDetectorコンポーネントはこのサンプルプロジェクトの中心です。それは、PSoC PLDに実装された設定可能な

5ビットの 2進数のシーケンス検出器です。 

4.2.1 SeqDetectorコンポーネントの作成手順 

SeqDetectorの作成手順はカウンターと同様です。 

1. Workplace Explorer の Components タブを選択します。MyLibrary プロジェクトを右クリックして、Add 

Component Itemをクリックします。 

2. Symbol Wizardコンポーネントのテンプレートを選択して、コンポーネントに SeqDetect_v1_00と名付けます。 

3. Create Newボタンをクリックして、コンポーネント シンボル ウィザードを起動します。 

4. 図 22のように、回路図シンボル用に 4個の入力端末と 2個の出力端末を定義します。 

図 22. SeqDetector用のシンボル作成ウィザード 

 

5. 図 23に示すように、OKをクリックして、シンボル回路図でシンボルを生成します。 

図 23. シーケンス検出器の最初のシンボル 

 

図 24に示すように、コンポーネントのサイズを変更できます。 

図 24. シーケンス検出器の最終のシンボル 

 

6. シンボル回路図で空きスペースを右クリックして、Propertiesをクリックします。 

プロパティ フィールドの Symbol セクションで、Catalog Placement の省略記号 (…) をクリックします。

CatalogPlacementで Community/Digital/Logic/Sequence Detector 5-bitを入力します。 

これにより、SeqDetectorは、Component Catalogの Communityタブで、「Digital」フォルダの「Logic」サブ

フォルダの中に配置されます。そのカタログ名は「Sequence Detector 5-bit」です。 

7. SeqDetectorのシーケンス値を設定可能にするために、コンポーネントのパラメーターを追加します。 

シンボル回路図で空きスペースを右クリックして、Symbol Parametersをクリックします。 

図 25のように、パラメーターの名称、タイプ、および初期値をそれぞれ「sequence」、「uint8」、および「22」

に指定します。 

図 25. SeqDetect用にパラメーターを定義する 

 

8. Parameter Definitionダイアログボックスの Miscセクションで Descriptionフィールドをクリックして、この

パラメーター用に記述します。 

9. Validators フィールドをクリックして、シーケンスの検証ツールを設定します。図 26 のように、シーケンス値

が 0～31の間にあることを保証するために検証ツールを設定します。OKをクリックして、変更を適用します。 

図 26. SeqDetect用の検証ツール 

 

10. Parameter Definitionダイアログ ボックスに戻った時、Hardwareフィールドを Trueにセットします。 

11. このためには、シンボル回路図で空きスペースを右クリックして、Generate Verilogをクリックします。 

次のステップとして、回路図シンボルを Verilogファイルにリンクさせます。 

12. Generate Verilogダイアログ ボックスですべての設定を初期値のままにして、Generateをクリックします。 

シーケンス検出器モジュール用の完全な Verilogコードは、付録 Cに記載されています。次の節では、コードの主な

パーツを説明します。 

4.2.2 Verilog設計: SeqDetector 

カウンターと同様、最初の段階として、SeqDetectモジュールの端末リストに出力を登録します。 

 output reg detect, 

 output reg restart, 

シーケンス検出器の重要な要素は、合計 6状態あるステートマシンです (図 8)。それぞれの状態に対応したローカル 

パラメーターを作成するために、下記のコードを#start body コメントの直後に加えます。 

 localparam START   = 3'd0; 

 localparam STATE_1 = 3'd1;  

 localparam STATE_2 = 3'd2;  

 localparam STATE_3 = 3'd3; 

 localparam STATE_4 = 3'd4;  

 localparam DETECT  = 3'd5; 

状態定義の定数が localparam キーワードを使って宣言されることに注意してください。これにより、他のモ

ジュールで同じ名前の定数と衝突することを防ぎます。 

状態変数を「register」タイプとして、パターン変数を「wire」タイプとして宣言します。 

reg [2:0] state_curr, state_next; 

wire [4:0] pattern = sequence; 

シーケンス検出器モジュールには、順次ブロックと組み合わせブロックという 2個の alwaysブロックがあります。 

always 順次ブロックはポジティブ エッジでトリガーされたフリップ フロップと同じように動作します。 

always 組み合わせブロックは、以下の通りに定義されるセンシティビティ リストがあります。 

always @ (oneIn or zeroIn or state_curr or pattern) 

注: PSoC Creator用に Verilog、always の文言はセンシティビティ リストを持たなければなりません。 

always 組み合わせブロックは、入力をデコードし、入力と現時点での状態に基づいて次の状態を割り当てます。1

か 0の入力のいずれかが検出されると、その入力がチェックされ、あるいは現時点での状態が維持されます。 

例えば、最初の状態は以下のようです。 

if((oneIn & pattern[4]) || 

(zeroIn & !pattern[4])) 

begin 

  state_next <= STATE_1;   

end 

else       

begin  

  state_next <= START; 

end 

注: pattern[4]はシーケンスの最初の正しい値を保存します。 

他の状態は同様に、コンポーネント入力を pattern[3], … , pattern[0]と比較して、次の状態をデコードします。 

Verilog ファイルへの変更が終わったら、保存します。この時点で、シーケンス検出器の設計は完了しました。それ

を設計で使用するには、カウンターの節で説明した手順に従います。 

5 データパス設計対 PLDベースの設計 

通信、タイミング、および制御アプリケーションは、各機能を支える論理構成に関する要求が異なります。 

経験則として、UDBリソースを利用する最適な方法は以下の通りです。 

▪ PLD (ランダム ロジック): 制御機能、CPLD統合、グルーロジック。 

▪ データパス (ストラクチャード ロジック): 通信、タイミング、演算。 

例えば、以下の、データパス対 PLDで実行された 8ビット演算と論理演算を考えてみます。 

機能 

PLDのみにおけるリソース消費 データパスのみにおけるリソース消費 

PLD 使用量 (%) データパス 使用量 (%) 

ADD8 5 10.4% 1 4.2% 

SUB8 5 10.4% 1 4.2% 

CMP8 3 6.3% 1 4.2% 

SHIFT8 3 6.3% 1 4.2% 

注:率は 24個の UDBを備えているデバイスで計算されます。 

PSoC PLDで複合機能を実行できますが、データパス モジュールを活用しないと、リソース不足になりがちです。 

6 まとめ 

このアプリケーション ノートでは、UDB PLDを紹介し、PSoC Creatorで Verilogベースのコンポーネントを作成す

るための設計プロセスについて説明します。このアプリケーション ノートをお読みいただいた後、PLD アーキテク

チャに詳しくなり、Verilogベースの独自のカスタム コンポーネントを作成することができます。 

PSoC UDB は、デジタル設計に柔軟かつ効果的なアーキテクチャを提供しています。簡単なものからある程度複雑

なものまで広範囲のロジック設計を PSoCの PLDに移植することができます。高度に複雑な設計は、PLDとデータ

パスを組み合わせて使用することで最もよく設計できます。UDB データパスの詳細については、AN82156 を参照し

てください。 

6.1 その他の情報 

付録 Aは、リソース カウントに関する、PSoC PLDと競合他社の CPLDとの比較表を記載します。 

付録 Bでは、2つの例を使ってマクロセルを介したデータフローを説明します。 

付録 Cには、シーケンス検出器モジュール用の完全な Verilogコードを記載します。 

付録 Dでは、プロジェクトの報告ファイルと静的タイミング解析を簡単に説明します。 

7 関連リソース 

アプリケーション ノート 

AN82156 - PSoC 3, PSoC 4 and PSoC 5LP Designing PSoC Creator Components with UDB Datapaths  

AN81623 – PSoC 3 and PSoC 5LP Digital Design Best Practices 

AN62510 – Implementing State Machines with PSoC 3 and PSoC 5LP 

AN61290 – PSoC 3 and PSoC 5LP Hardware Design Considerations 

AN72382 – Using PSoC 3 and PSoC 5LP GPIO Pins 

AN60580 – SIO Tips and Tricks in PSoC 3 and PSoC 5LP 

AN54181 – Getting started with PSoC 3 

AN79953 – Getting Started with PSoC 4 

AN77759 – Getting started with PSoC 5LP 

AN221774 Getting Started with PSoC 6 MCUs 

KB記事 

KBA86336 – Just Enough Verilog for PSoC 

KBA86338 – Creating a Verilog-based Component 

KBA81772 – Adding Component Primitives / Verilog Components to a Project 

Basics of Verilog and Datapath Configuration Tool for Component Creation  

ビデオ 

以下のビデオは、PSoC Creatorおよび Verilogコンポーネントの作成プロセスについて説明します。 

基本 

Creating a New Project 

Using the Start Page 

コンポーネントの作成 

PSoC Creator 113: PLD Based Verilog Components  

Creating a New Component Symbol  

Creating a Verilog Implementation 

Creating a Schematic Implementation 

A 付録 A: PSoC PLDと競合他社の CPLDとのリソース比較 
表 1では、PSoC PLDリソースを競合他社の同じサイズの CPLDと比較しています。この表では、UDBデータパス

でのプログラマブル ロジックを考慮に入れていないことに注意してください。PSoC PLD とデータパスの両方を使

用すれば、PSoCはより大きなサイズの CPLDと競争できるようになります。 

表 1. PSoC PLDと競合他社の PLDとのマクロセル比較 

デバイス 
マクロセル 

(MC) 
ブロック数 

ブロックごとの マ

クロセル (MC) 数 

ブロックごと

の入力数 

プロダクト 

ターム (PT) 数 

ブロックごとのプロ

ダクト ターム (PT) 数 

サイプレス PSoC        

スーパセット PSoC 3、
PSoC 5LP 

192 48 4 12 384 8 

CY8C42 32 8 4 12 64 8 

Altera MAX-II       

EPM240 128～240* 24 10 36 * * 

Lattice ispMACH       

4032ZE 32 2 16 36 160 80 

4064ZE 64 4 16 36 320 80 

40128ZE 128 8 16 36 640 80 

Xilinx Coolrunner-II       

XC2C32A 32 2 16 56 112 56 

XC2C64A 64 4 16 56 224 56 

XC2C128 128 8 16 56 448 56 

* Altera MAX-IIは従来のプロダクト タームのアーキテクチャではありません。 

B 付録 B: マクロセル構成図 
図 27および図 28には、それぞれ Dフリップ フロップ (D-FF) と Tフリップ フロップ (T-FF) 機能のマクロセルを介

したデータ フローを示します。 

図 27. D-FF機能が有効なマクロセル 

 

図 28. T-FF機能が有効なマクロセル 

C 付録 C: シーケンス検出器の Verilogコード 

module SeqDetect_v1_20 ( 

 output reg detect, 

 output reg restart, 

 input   clock, 

 input   oneIn, 

 input   reset, 

 input   zeroIn 

); 

 /* Note that the value assigned to the parameter in this line 

  * has no effect. The actual parameter value is taken from  

  * the component customizer. 

  */ 

 parameter sequence = 0; 

 

//`#start body` -- edit after this line, do not edit this line 

 /* Six states are required. 

  * The states follow START -> STATE_1 -> ... -> DETECT if the  

  * correct inputs are entered. As soon as a wrong input is entered 

  * the design jumps to the START state. The states are defined as  

  * localparams (instead of `defines) to limit their scope to this  

  * module only.  

  */  

 localparam START   = 3'd0;  /* detect, restart = 0, 1 */ 

 localparam STATE_1 = 3'd1;  /* detect, restart = 0, 0 */ 

 localparam STATE_2 = 3'd2;  /* detect, restart = 0, 0 */ 

 localparam STATE_3 = 3'd3;  /* detect, restart = 0, 0 */ 

 localparam STATE_4 = 3'd4;  /* detect, restart = 0, 0 */ 

 localparam DETECT  = 3'd5;  /* detect, restart = 1, 0 */ 

  

/* registered value to hold 3-bit state */ 

reg [2:0] state_curr, state_next; 

  

 /* pattern[4:0] holds the user-supplied sequence value  

  * suppose sequence = 22 then pattern[4:0] = 5'b10110 

  * Note that pattern[4] is the first-entered user input 

  */ 

 wire [4:0] pattern = sequence; 

 

  /* Sequential block of the state machine - outputs are assigned here */ 

 always @ (posedge clock) 

 begin 

  /* reset causes the component to enter the START state */ 

  if(reset) 

  begin  

   state_curr <= START; 

   /* Immediately assign detect and restart values */ 

   detect <= 1'b0; 

   restart <= 1'b1; 

  end 

  else /* reset is not asserted - go through states */ 

  begin 

   state_curr <= state_next; 

    

   /* Assign 'detect' value - 1 only in DETECT state, 0 otherwise */ 

   if (state_next == DETECT)  

   begin 

    detect <= 1'b1; 

   end 

   else 

   begin 

    detect <= 1'b0; 

   end 

    

   /* Assign 'restart' value - 1 only in RESTART state, 0 otherwise*/ 

   if (state_next == START)  

   begin 

    restart <= 1'b1; 

   end 

   else 

   begin 

    restart <= 1'b0; 

   end 

  end 

 end 

  

 /* Finite State Machine combinatorial block - contains most of the  

  * combinatorial logic. 

  */ 

 always @ (oneIn or zeroIn or state_curr or pattern)  

 begin 

  /* If either a one or zero has been entered, take action */ 

  if(oneIn | zeroIn) 

  begin 

   case(state_curr) 

    START:  /* Initial state */ 

    begin 

     /* check whether the first bit entered is correct */ 

     if((oneIn & pattern[4]) || (zeroIn & !pattern[4])) 

     begin 

     state_next <= STATE_1;/* advance to the next state */ 

     end 

     else    /* revert to the initial state */ 

     begin  

      state_next <= START; 

     end 

    end 

     

    STATE_1: /* First input is correct */ 

    begin 

     if((oneIn & pattern[3]) || (zeroIn & !pattern[3])) 

     begin 

      state_next <= STATE_2; 

     end 

     else 

     begin 

      state_next <= START; 

     end 

    end     

     

    STATE_2: /* Two inputs are correct */ 

    begin 

     if((oneIn & pattern[2]) || (zeroIn & !pattern[2])) 

     begin  

      state_next <= STATE_3; 

     end 

     else 

     begin 

      state_next <= START; 

     end 

    end     

    STATE_3: /* Three inputs are correct */ 

    begin 

     if((oneIn & pattern[1]) || (zeroIn & !pattern[1])) 

     begin  

      state_next <= STATE_4; 

     end 

     else 

     begin 

      state_next <= START; 

     end 

    end   

     

    STATE_4:  /* Four inputs are correct */ 

    begin 

     if((oneIn & pattern[0]) || (zeroIn & !pattern[0])) 

     begin 

      state_next <= DETECT; 

     end 

     else 

     begin 

      state_next <= START; 

     end 

    end 

     

    DETECT:  /* All five inputs are correct! */ 

    begin 

/* When in the detect state, if an input is given, show same behavior as START */ 

/* check whether the bit entered is the correct beginning to a new sequence*/ 

     if((oneIn & pattern[4]) || (zeroIn & !pattern[4])) 

     begin  

      state_next <= STATE_1; 

     end 

     else  /* revert to the initial state */ 

     begin  

      state_next <= START; 

     end 

    end 

     

    default:/* we should never get here - reset the component*/ 

    begin 

     state_next <= START; 

    end 

   endcase 

  end 

  else  /* if neither 1 or 0 have been entered, stay in same state */ 

  begin 

   state_next <= state_curr; 

  end 

 end 

  

//`#end` -- edit above this line, do not edit this line 

endmodule  

D 付録 D: ビルド後の設計の注意事項 

D.1 プロジェクトの報告ファイル 

Workspace Explorerウィンドウの Resultsタブからプロジェクトのビルド レポート (<project_name>.rpt) をアクセ

スします。正常に完了したビルド後に作成されます。以下は、報告ファイルの主な部分です。 

▪ 技術的なマッピングのまとめ: マクロセル、プロダクト ターム、データパス、ピン、クロック分周器などが図 29

に示されます。 

図 29. PSoC Creatorプロジェクトのビルド レポート ファイル 

 

▪ 合成結果: Verilog コンパイル、構文解析、高レベルの合成、最適化など合成の各段階で生成されたエラーと警告

を一覧表示します。同節は、シンセサイザによって最適化して除去したロジックの詳細情報を含みます。これは、

デバッギングおよびトラブルシューティングに役立ちます。 

▪ デジタル部の配置: PLDパッキングのまとめ。図 30は PLDの利用例を示します。 

図 30. PLDパッキングのまとめレポート 

 

▪ デジタル部の配置: PLDパッキングのまとめ: PLDの統計値: 図 31は、ロジック アレイ ブロック (LAB) ごとの

PLD PTとマクロセルの平均使用量の例を示します。 

図 31. PLDの使用量のレポート例 

 

▪ 最終配置のまとめ: コンポーネントの詳細を提示します。同節は、UDB の使用量、占有率、統計値、および配置 

(座標) の詳細を示します。 

D.2 静的タイミング解析 

デジタル設計のデバッグ処理の重要な一部は、静的タイミング解析 (STA) です。STA はデジタル設計を評価し、信

号の入出力間の遅延を計算します。それらの遅延に基づいて、設計で使用される各クロックの最大許容周波数を算出

します。 

プロジェクトをビルドすると、PSoC Creator は静的タイミング解析レポートを自動的に作成します。このレポート

は、各クロックの周波数を制限する、設計内の重要な部分を提示します。算出された最大周波数が要求されたクロッ

ク周波数より小さい場合、設計内にタイミング違反が存在することを示す警告を表示します。 

タイミング違反の回避方法、および PSoC Creator STA 警告の処理の詳細については、「AN81623 – PSoC 3 and 

PSoC 5LP Digital Design Best Practices」を参照してください。 

改訂履歴 

文書名: AN82250 - PSoC Creatorの Verilogによるプログラマブル ロジック設計の実装方法 

文書番号: 001-89367 

版 ECN 発行日 変更内容 

** 4136657 09/26/2013 これは英語版 001-82250 Rev. *D を翻訳した日本語版 001-89367 Rev. **です。 

*A 4722751 04/27/2015 これは英語版 001-82250 Rev. *F を翻訳した日本語版 001-89367 Rev. *A です。 

*B 5184202 03/22/2016 これは英語版 001-82250 Rev. *Hを翻訳した日本語版 001-89367 Rev. *Bです。 

*C 5473611 10/13/2016 新しいテンプレートに更新しました。 

サンセットレビューを完了する。 

*D 5817630 07/19/2017 ロゴと著作権を更新しました。 

*E 6504619 08/27/2019 これは英語版 001-82250 Rev. *Jを翻訳した日本語版 001-89367 Rev. *Eです。 

